

# NAG Library Routine Document

## F08JEF (DSTEQR)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F08JEF (DSTEQR) computes all the eigenvalues and, optionally, all the eigenvectors of a real symmetric tridiagonal matrix, or of a real symmetric matrix which has been reduced to tridiagonal form.

### 2 Specification

```
SUBROUTINE F08JEF (COMPZ, N, D, E, Z, LDZ, WORK, INFO)
  INTEGER          N, LDZ, INFO
  REAL (KIND=nag_wp) D(*), E(*), Z(LDZ,*), WORK(*)
  CHARACTER(1)    COMPZ
```

The routine may be called by its LAPACK name *dsteqr*.

### 3 Description

F08JEF (DSTEQR) computes all the eigenvalues and, optionally, all the eigenvectors of a real symmetric tridiagonal matrix  $T$ . In other words, it can compute the spectral factorization of  $T$  as

$$T = ZAZ^T,$$

where  $A$  is a diagonal matrix whose diagonal elements are the eigenvalues  $\lambda_i$ , and  $Z$  is the orthogonal matrix whose columns are the eigenvectors  $z_i$ . Thus

$$Tz_i = \lambda_i z_i, \quad i = 1, 2, \dots, n.$$

The routine may also be used to compute all the eigenvalues and eigenvectors of a real symmetric matrix  $A$  which has been reduced to tridiagonal form  $T$ :

$$\begin{aligned} A &= QTQ^T, \text{ where } Q \text{ is orthogonal} \\ &= (QZ)A(QZ)^T. \end{aligned}$$

In this case, the matrix  $Q$  must be formed explicitly and passed to F08JEF (DSTEQR), which must be called with  $\text{COMPZ} = 'V'$ . The routines which must be called to perform the reduction to tridiagonal form and form  $Q$  are:

|                             |  |
|-----------------------------|--|
| full matrix                 | F08FEF (DSYTRD) and F08FFF (DORGTR)        |
| full matrix, packed storage | F08GEF (DSPTRD) and F08GFF (DOPGTR)        |
| band matrix                 | F08HEF (DSBTRD) with $\text{VECT} = 'V'$ . |

F08JEF (DSTEQR) uses the implicitly shifted  $QR$  algorithm, switching between the  $QR$  and  $QL$  variants in order to handle graded matrices effectively (see Greenbaum and Dongarra (1980)). The eigenvectors are normalized so that  $\|z_i\|_2 = 1$ , but are determined only to within a factor  $\pm 1$ .

If only the eigenvalues of  $T$  are required, it is more efficient to call F08JFF (DSTERF) instead. If  $T$  is positive definite, small eigenvalues can be computed more accurately by F08JGF (DPTEQR).

## 4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Greenbaum A and Dongarra J J (1980) Experiments with QR/QL methods for the symmetric triangular eigenproblem *LAPACK Working Note No. 17 (Technical Report CS-89-92)* University of Tennessee, Knoxville <http://www.netlib.org/lapack/lawnspdf/lawn17.pdf>

Parlett B N (1998) *The Symmetric Eigenvalue Problem* SIAM, Philadelphia

## 5 Arguments

- 1: COMPZ – CHARACTER(1) *Input*  
*On entry:* indicates whether the eigenvectors are to be computed.  
 COMPZ = 'N'  
     Only the eigenvalues are computed (and the array Z is not referenced).  
 COMPZ = 'V'  
     The eigenvalues and eigenvectors of  $A$  are computed (and the array Z must contain the matrix  $Q$  on entry).  
 COMPZ = 'I'  
     The eigenvalues and eigenvectors of  $T$  are computed (and the array Z is initialized by the routine).  
*Constraint:* COMPZ = 'N', 'V' or 'I'.
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $T$ .  
*Constraint:*  $N \geq 0$ .
- 3: D(\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the dimension of the array D must be at least  $\max(1, N)$ .  
*On entry:* the diagonal elements of the tridiagonal matrix  $T$ .  
*On exit:* the  $n$  eigenvalues in ascending order, unless INFO > 0 (in which case see Section 6).
- 4: E(\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the dimension of the array E must be at least  $\max(1, N - 1)$ .  
*On entry:* the off-diagonal elements of the tridiagonal matrix  $T$ .  
*On exit:* E is overwritten.
- 5: Z(LDZ, \*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array Z must be at least  $\max(1, N)$  if COMPZ = 'V' or 'I' and at least 1 if COMPZ = 'N'.  
*On entry:* if COMPZ = 'V', Z must contain the orthogonal matrix  $Q$  from the reduction to tridiagonal form.  
 If COMPZ = 'I', Z need not be set.  
*On exit:* if COMPZ = 'V' or 'I', the  $n$  required orthonormal eigenvectors stored as columns of Z; the  $i$ th column corresponds to the  $i$ th eigenvalue, where  $i = 1, 2, \dots, n$ , unless INFO > 0.  
 If COMPZ = 'N', Z is not referenced.

- 6: LDZ – INTEGER *Input*  
*On entry:* the first dimension of the array Z as declared in the (sub)program from which F08JEF (DSTEQR) is called.  
*Constraints:*  
 if COMPZ = 'V' or 'I', LDZ  $\geq$  max(1, N);  
 if COMPZ = 'N', LDZ  $\geq$  1.
- 7: WORK(\*) – REAL (KIND=nag\_wp) array *Workspace*  
**Note:** the dimension of the array WORK must be at least  $\max(1, 2 \times (N - 1))$  if COMPZ = 'V' or 'I' and at least 1 if COMPZ = 'N'.  
 If COMPZ = 'N', WORK is not referenced.
- 8: INFO – INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The algorithm has failed to find all the eigenvalues after a total of  $30 \times N$  iterations. In this case, D and E contain on exit the diagonal and off-diagonal elements, respectively, of a tridiagonal matrix orthogonally similar to  $T$ . If INFO =  $i$ , then  $i$  off-diagonal elements have not converged to zero.

## 7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix  $(T + E)$ , where

$$\|E\|_2 = O(\epsilon)\|T\|_2,$$

and  $\epsilon$  is the *machine precision*.

If  $\lambda_i$  is an exact eigenvalue and  $\tilde{\lambda}_i$  is the corresponding computed value, then

$$|\tilde{\lambda}_i - \lambda_i| \leq c(n)\epsilon\|T\|_2,$$

where  $c(n)$  is a modestly increasing function of  $n$ .

If  $z_i$  is the corresponding exact eigenvector, and  $\tilde{z}_i$  is the corresponding computed eigenvector, then the angle  $\theta(\tilde{z}_i, z_i)$  between them is bounded as follows:

$$\theta(\tilde{z}_i, z_i) \leq \frac{c(n)\epsilon\|T\|_2}{\min_{i \neq j} |\lambda_i - \lambda_j|}.$$

Thus the accuracy of a computed eigenvector depends on the gap between its eigenvalue and all the other eigenvalues.

## 8 Parallelism and Performance

F08JEF (DSTEQR) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08JEF (DSTEQR) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of floating-point operations is typically about  $24n^2$  if COMPZ = 'N' and about  $7n^3$  if COMPZ = 'V' or 'I', but depends on how rapidly the algorithm converges. When COMPZ = 'N', the operations are all performed in scalar mode; the additional operations to compute the eigenvectors when COMPZ = 'V' or 'I' can be vectorized and on some machines may be performed much faster.

The complex analogue of this routine is F08JSF (ZSTEQR).

## 10 Example

This example computes all the eigenvalues and eigenvectors of the symmetric tridiagonal matrix  $T$ , where

$$T = \begin{pmatrix} -6.99 & -0.44 & 0.00 & 0.00 \\ -0.44 & 7.92 & -2.63 & 0.00 \\ 0.00 & -2.63 & 2.34 & -1.18 \\ 0.00 & 0.00 & -1.18 & 0.32 \end{pmatrix}.$$

See also the examples for F08FFF (DORGTR), F08GFF (DOPGTR) or F08HEF (DSBTRD), which illustrate the use of this routine to compute the eigenvalues and eigenvectors of a full or band symmetric matrix.

### 10.1 Program Text

```

Program f08jefe

!      F08JEF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: dsteqr, nag_wp, x04caf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, info, ldz, n
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: d(:), e(:), work(:), z(:, :)
!      .. Executable Statements ..
      Write (nout,*) 'F08JEF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      ldz = n
      Allocate (d(n),e(n-1),work(2*n-2),z(ldz,n))

!      Read T from data file

      Read (nin,*) d(1:n)
      Read (nin,*) e(1:n-1)

!      Calculate all the eigenvalues and eigenvectors of T
!      The NAG name equivalent of dsteqr is f08jef
      Call dsteqr('I',n,d,e,z,ldz,work,info)

```

```

Write (nout,*)
If (info>0) Then
  Write (nout,*) 'Failure to converge.'
Else
!      Print eigenvalues and eigenvectors

  Write (nout,*) 'Eigenvalues'
  Write (nout,99999) d(1:n)
  Write (nout,*)
  Flush (nout)

!      Standardize the eigenvectors so that first elements are non-negative.
  Do i = 1, n
    If (z(1,i)<0.0_nag_wp) Then
      z(1:n,i) = -z(1:n,i)
    End If
  End Do

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
  ifail = 0
  Call x04caf('General',' ',n,n,z,ldz,'Eigenvectors',ifail)

  End If

99999 Format (3X,(8F8.4))
End Program f08jefe

```

## 10.2 Program Data

```

F08JEF Example Program Data
  4      :Value of N
-6.99   7.92   2.34   0.32
-0.44  -2.63  -1.18      :End of matrix T

```

## 10.3 Program Results

F08JEF Example Program Results

```

Eigenvalues
  -7.0037 -0.4059  2.0028  8.9968

```

```

Eigenvectors
      1      2      3      4
1  0.9995  0.0109  0.0167  0.0255
2  0.0310 -0.1627 -0.3408 -0.9254
3  0.0089 -0.5170 -0.7696  0.3746
4  0.0014 -0.8403  0.5397 -0.0509

```

---